

Integrantes: Arthur dos Santos Gomes, Daniel Lopes Tayt-Sohn Pismo, João Pedro B. Dias, Kauann Rimes Pinto, Nathan dos Santos Gomes, Paulo Ricardo Correa Cardoso

Questão 1: <https://portugol.dev/#share=3bjbrzt>

```
programa {  
    funcao inicio() {  
        inteiro numero=0  
        escreva("Entre com um número natural: ")  
        leia(numero)  
  
        escreva("Divisores do número:" ,numero, "\n")  
        para(inteiro divisor=1; divisor<=numero;divisor++){  
            se(numero%divisor==0){  
                escreva( " ", divisor)  
  
            }  
  
        }  
  
    }  
}
```

Questão 2: <https://portugol.dev/#share=ccee4yd>

```
programa
{
    inclua biblioteca Texto --> txt

    funcao inicio()
    {
        cadeia texto
        inteiro contador_vogais = 0
        inteiro i
        caracter caractere_atual

        escreva("Digite um texto: ")
        leia(texto)

        para (i = 0; i < txt.numero_caracteres(texto); i++)
        {
            caractere_atual = txt.obter_caracter(texto, i)
            se (eh_vogal(caractere_atual))
            {
                contador_vogais++
            }
        }

        escreva("O texto possui ", contador_vogais, " vogais.\n")
    }

    funcao logico eh_vogal(caracter caractere)
    {
        cadeia vogais = "aeiouAEIOU"

        para (inteiro i = 0; i < txt.numero_caracteres(vogais); i++)
        {
            se (caractere == txt.obter_caracter(vogais, i))
            {
                retorne verdadeiro
            }
        }

        retorne falso
    }
}
```

Questão 3: <https://portugol.dev/#share=e1mbzkl>

```
programa{
    funcao inicio()
    {
        inteiro numeros[10]
        inteiro pares[10]
        inteiro impares[10]
        inteiro contador_pares = 0
        inteiro contador_impares = 0
        inteiro i
        para (i = 0; i < 10; i++)
        {
            escreva("Digite o ", i + 1, "º número: ")
            leia(numeros[i])
        }
        escreva("Números pares:\n")
        para (i = 0; i < 10; i++)
        {
            se (numeros[i] % 2 == 0)
            {
                pares[contador_pares] = numeros[i]
                escreva(pares[contador_pares], " ")
                contador_pares++
            }
        }
        escreva("\nNúmeros ímpares:\n")
        para (i = 0; i < 10; i++)
        {
            se (numeros[i] % 2 != 0)
            {
                impares[contador_impares] = numeros[i]
                escreva(impares[contador_impares], " ")
                contador_impares++
            }
        }
        escreva("\n")
    }
}
```

Questão 4: <https://portugol.dev/#share=c1zpeunl>

```
programa {  
  funcao inicio() {  
    inteiro base , expoente, pot, i  
  
    escreva("Escreva a base: ")  
    leia(base)  
    pot = base  
    limpa()  
    escreva("Escreva o expoente (ex: ³): ")  
    leia(expoente)  
    limpa()  
    i = 0  
    faca {  
      pot = pot*base  
      escreva(pot)  
      i++  
    }  
    enquanto (i < expoente)  
  }  
}
```

Questão 5: <https://portugol.dev/#share=4yrp9ze>

```
programa
{

    funcao inicio()
    {
        inteiro V[10], i , j, aux

        escreva("Informe 10 números:\n")
        para (i = 0; i <= 9; i = i + 1) {
            leia(V[i])
            limpa()
            escreva("Informe 10 números:\n")
        }

        para(i = 0; i <= 9; i = i + 1) {
            para(j = i +1; j <= 9; j = j + 1) {
                se (V[j] < V[i]) {
                    aux = V[j]
                    V[j] = V[i]
                    V[i] = aux
                }
            }
        }
        limpa()
        escreva("Sequência ordenada: ")
        para(i = 0; i <= 9; i = i + 1) {
            escreva(V[i], " , ")
        }
    }
}
```

Questão 6: <https://portugol.dev/#share=voyw97l>

programa

```
{
    funcao inicio()
    {
        inteiro matriz[3][3], linha, coluna
        logico identidade = verdadeiro

        para (linha = 0; linha < 3; linha++)
        {
            para (coluna = 0; coluna < 3; coluna++)
            {
                escreva("Elemento [", linha, "][", coluna, "]: ")
                leia(matriz[linha][coluna])
            }
        }
        para (linha = 0; linha < 3; linha++)
        {
            para (coluna = 0; coluna < 3; coluna++)
            {
                se (linha == coluna e matriz[linha][coluna] != 1)
                {
                    identidade = falso
                }
                senao se (linha != coluna e matriz[linha][coluna] != 0)
                {
                    identidade = falso
                }
            }
        }
        se (identidade == verdadeiro)
        {
            escreva("A matriz ĄŠ identidade.\n")
        }
        senao
        {
            escreva("A matriz nŁo ĄŠ identidade.\n")
        }
    }
}
```

Questão 7: <https://portugol.dev/#share=0xcmmfk>

```
programa
{
    funcao inicio()
    {
        inteiro numero, numeroOriginal, numeroInvertido = 0, resto

        escreva("Digite um número inteiro: ")
        leia(numero)

        numeroOriginal = numero

        enquanto (numero > 0)
        {
            resto = numero % 10
            numeroInvertido = numeroInvertido * 10 + resto
            numero = numero / 10
        }

        se (numeroOriginal == numeroInvertido)
        {
            escreva("É palíndromo.\n")
        }
        senao
        {
            escreva("Não é palíndromo.\n")
        }
    }
}
```

Questão 8: <https://portugol.dev/#share=au6pkh3>

```
programa
{

    funcao inicio()
    {
        inteiro numX, valor = 1, divisores = 0
        escreva("Digite um número: ")
        leia(numX)

        se(numX > 0) {
            enquanto(valor <= numX){
                se(numX % valor == 0){
                    divisores = divisores + 1
                }

                valor = valor + 1
            }
            se(divisores == 2) {
                escreva("O número " + numX + " é primo.")
            }
            senao{
                escreva("O número " + numX + " não é primo.")
            }
        }

        senao{
            escreva("Valor inválido")
        }
    }
}
```



Questão 9: <https://portugol.dev/#share=7yj6uwt>

programa

{

inclua biblioteca Util-->util

funcao inicio() {

inteiro quantidadeDeTermos, termo1 = 0, termo2 = 1

inteiro proximoTermo

escreva("Digite a quantidades de termos: ")

leia(quantidadeDeTermos)

limpa()

escreva("A sequência de Fibonacci com " + quantidadeDeTermos + " termos é: \n")

escreva(termo1, " " + termo2 + " ")

util.aguarde(270)

para(inteiro i = 2; i <= quantidadeDeTermos; i++) {

proximoTermo = termo1 + termo2

termo1 = termo2

termo2 = proximoTermo

util.aguarde(270)

escreva(proximoTermo, " ")

}

}

}

Questão 10: <https://portugol.dev/#share=2bvkafe>

programa {

inclua biblioteca Texto --> txt

funcao inicio() {  
cadeia palavraPedida, palavraInvertido

escreva("Digite uma palavra para fazer a inversão: \n")  
leia(palavraPedida)  
palavraInvertido = inverterCadeia(palavraPedida)  
escreva("Essa é a palavra invertida: " + palavraInvertido)

}  
funcao cadeia inverterCadeia(cadeia texto){  
inteiro tamanho = txt.numero\_caracteres(texto)  
se(tamanho <= 1){

retorne texto

} senao{

cadeia ultimaLetra = txt.extrair\_subtexto(texto, tamanho - 1, tamanho)

cadeia resto = txt.extrair\_subtexto(texto, 0 , tamanho - 1)

retorne ultimaLetra + inverterCadeia(resto)

}  
}  
}

Questão 11: <https://portugol.dev/#share=ct6ojht>

```
programa
{

funcao inicio()
{
    inteiro vetor[10], somaPar = 0, somaImpar = 0, qtdPar = 0, qtdImpar = 0
    real mediaPar, mediaImpar

    para(inteiro x = 0; x < 10; x++){
        escreva("Digite um número: ")
        leia(vetor[x])
        se(vetor[x] % 2 == 0){
            somaPar += vetor[x]
            qtdPar++

        }senao{
            somaImpar += vetor[x]
            qtdImpar++
        }
    }se(qtdPar > 0){
        mediaPar = somaPar/qtdPar
        escreva("\nMédia dos números pares: " + mediaPar)
    }senao{
        escreva("\nNenhum número par foi inserido")
    }

    se(qtdImpar > 0){
        mediaImpar = somaImpar/qtdImpar
        escreva("\nMédia dos números ímpares: " + mediaImpar)
    }senao{
        escreva("\nNenhum número ímpar foi inserido")
    }
    escreva("\nQuantidade de números usados Par: " + qtdPar)
    escreva("\nQuantidade de números usados Impar: " + qtdImpar)

}
}
```

Questão 12: <https://portugol.dev/#share=tmt6o5i>

programa

```
{  
  
    funcao inteiro fatorial(inteiro x){  
        se(x == 0 ou x == 1)  
            retorne 1  
        senao{  
            retorne x * fatorial(x - 1)  
        }  
    }  
    funcao inicio()  
    {  
        inteiro valor  
  
        escreva("Digite um valor número: ")  
        leia(valor)  
  
        escreva("\nFatorial de ", valor, "é: ", fatorial(valor))  
    }  
}
```

Questão 13: <https://portugol.dev/#share=l3qclmc>

```
programa
{
    funcao inicio()
    {
        inteiro matriz[4][4], linha, coluna, maiorValor, menorValor
        para (linha = 0; linha < 4; linha++)
        {
            para (coluna = 0; coluna < 4; coluna++)
            {
                escreva("Elemento [", linha, "][", coluna, "]: ")
                leia(matriz[linha][coluna])
            }
        }
        maiorValor = matriz[0][0]
        menorValor = matriz[0][0]
        para (linha = 0; linha < 4; linha++)
        {
            para (coluna = 0; coluna < 4; coluna++)
            {
                se (matriz[linha][coluna] > maiorValor)
                {
                    maiorValor = matriz[linha][coluna]
                }
                se (matriz[linha][coluna] < menorValor)
                {
                    menorValor = matriz[linha][coluna]
                }
            }
        }
        escreva("\nMaior valor: ", maiorValor, "\n")
        escreva("Menor valor: ", menorValor, "\n")
    }
}
```

Questão 14: <https://portugol.dev/#share=6iy0l0g>

```
programa
{
    funcao inicio()
    {
        inteiro altura, i,piramide

        escreva("Digite a altura da pirâmide: ")
        leia(altura)

        para (i = 1; i <= altura; i++)
        {
            para (piramide = 1; piramide <= i; piramide++)
            {
                escreva(piramide, " ")
            }
            escreva("\n")
        }
    }
}
```

Questão 15: <https://portugol.dev/#share=7lkv9bb>

programa

```
{
    funcao inicio()
    {
        inteiro primeiraColuna[5], segundaColuna[5], vetorSoma[5]
        inteiro i
        para (i = 0; i < 5; i++)
        {
            escreva("Digite o ", i + 1, "º numero da primeira coluna: ")
            leia(primeiraColuna[i])
        }
        para (i = 0; i < 5; i++)
        {
            escreva("Digite o ", i + 1, "º numero da segunda coluna: ")
            leia(segundaColuna[i])
        }
        para (i = 0; i < 5; i++)
        {
            vetorSoma[i] = primeiraColuna[i] + segundaColuna[i]
        }

        escreva("Soma dos respectivos numeros: ")
        para (i = 0; i < 5; i++)
        {
            escreva(vetorSoma[i], " ")
        }
    }
}
```

Questão 16: <https://portugol.dev/#share=oebp441>

programa

```
{
    funcao inicio()
    {
        inteiro numeros[10]
        inteiro negativos
        para (inteiro i = 0; i < 10; i++)
        {
            escreva("Digite o ", i + 1, "º número: ")
            leia(numeros[i])
        }
        negativos = contarNegativos(numeros)
        escreva("Quantidade de números negativos: ", negativos)
    }
    funcao inteiro contarNegativos(inteiro numeros[])
    {
        inteiro contador=0
        para (inteiro i = 0; i < 10; i++)
        {
            se (numeros[i] < 0)
            {
                contador++
            }
        }
        retorne contador
    }
}
```



Questão 17: <https://portugol.dev/#share=ceg4nhi>

```
programa
{
    funcao inicio()
    {
        real altura, somaHomens = 0.0, somaMulheres = 0.0
        inteiro Homens = 0, Mulheres = 0
        cadeia sexo
        inteiro i

        para (i = 0; i < 5; i++)
        {
            escreva("Digite a altura da ", i + 1, "ª pessoa (em metros): ")
            leia(altura)

            se (altura < 0.0)
            {
                retorne
            }

            escreva("Digite o sexo da ", i + 1, "ª pessoa (M/F): ")
            leia(sexo)

            se (sexo == "M" )
            {
                somaHomens += altura
                Homens++
            }
            senao se (sexo == "F" )
            {
                somaMulheres += altura
                Mulheres++
            }
            senao
            {
                escreva("Sexo inválido.\n")
                i--
            }
        }

        se (Homens > 0)
        {
            escreva("Média da altura dos homens: ", somaHomens / Homens, " metros.\n")
        }
        senao
        {
            escreva("Nenhum homem informado.\n")
        }

        se (Mulheres > 0)
        {
            escreva("Média da altura das mulheres: ", somaMulheres / Mulheres, " metros.\n")
        }
        senao
        {
            escreva("Nenhuma mulher informada.\n")
        }
    }
}
```

Questão 18: <https://portugol.dev/#share=qwia51i>

```
programa{
    funcao inicio()
    {
        inteiro matriz[4][4]
        inteiro linha, coluna, soma_diagonal = 0
        para (linha = 0; linha < 4; linha++)
        {
            para (coluna = 0; coluna < 4; coluna++)
            {
                escreva("Valor [", linha, "][", coluna, "]: ")
                leia(matriz[linha][coluna])
            }
        }
        para (linha = 0; linha < 4; linha++)
        {
            soma_diagonal = soma_diagonal + matriz[linha][3 - linha]
        }
        escreva("\nSoma da diagonal secundária: ", soma_diagonal, "\n")
    }
}
```

Questão 19: <https://portugol.dev/#share=rny7834>

```
programa {  
    funcao inicio() {  
        inteiro numeroEntrada  
        inteiro contador = 0  
  
        escreva("Digite um número: ")  
        leia(numeroEntrada)  
  
        faca {  
            contador++  
            escreva("Digite um número: ")  
            leia(numeroEntrada)  
        } enquanto (numeroEntrada >= 0)  
  
        limpa()  
        escreva("Quantidade de números lidos: ", contador)  
    }  
}
```

Questão 20: <https://portugol.dev/#share=cs8j78s>

```
programa {
    funcao inicio() {
        cadeia usuario
        cadeia senhaEntrada
        cadeia senhaCorreta = "ABC123"
        inteiro i

        escreva("Bem-vindo, digite seu nome de usuário: ")
        leia(usuario)
        limpa()

        para(i = 3; i > 0; i--) {
            escreva("Digite sua senha: ")
            leia(senhaEntrada)
            limpa()

            se(senhaEntrada == senhaCorreta) {
                escreva("Acesso Concedido!")
                pare
            } senao {
                escreva("Senha Incorreta. Tentativas Restantes: " + (i - 1) + "\n")
            }
        }

        se(i == 0) {
            escreva("Você excedeu o número de tentativas. Acesso Bloqueado.")
        }
    }
}
```

Questão 21: <https://portugol.dev/#share=cs8j78s>

```
programa
{
    funcao inicio()
    {
        inteiro matriz[5][5], linha, coluna, somaLinha, maiorSoma = -99999, linhaMaiorSoma =
-1
        para (linha = 0; linha < 5; linha++)
        {
            para (coluna = 0; coluna < 5; coluna++)
            {
                escreva("Valor [", linha, "][", coluna, "]: ")
                leia(matriz[linha][coluna])
            }
        }
        para (linha = 0; linha < 5; linha++)
        {
            somaLinha = 0
            para (coluna = 0; coluna < 5; coluna++)
            {
                somaLinha = somaLinha + matriz[linha][coluna]
            }

            se (somaLinha > maiorSoma)
            {
                maiorSoma = somaLinha
                linhaMaiorSoma = linha + 1
            }
        }
        escreva("\nA linha com a maior soma é a linha ", linhaMaiorSoma, " com soma ",
maiorSoma, ".\n")
    }
}
```